Notes on the call-bys

Call-by-value

```
(define foo-by-value
   (lambda (x)
     (bar x)
     x))

(define bar-by-value
  (lambda (y)
    (set! y 10)))

> (foo-by-value 5)
5
```

This is another version of call-by-value using boxes:

```
(define foo-by-value/boxes
   (lambda (x)
     (bar-by-value/boxes (box (unbox x)))
     (unbox x)))

(define bar-by-value/boxes
  (lambda (y)
    (set-box! y 10)))

> (foo-by-value/boxes (box 5))
5
```

Call-by-reference

```
(define foo-by-ref
  (lambda (x)
    (bar-by-ref x)
    (unbox x)))

(define bar-by-ref
  (lambda (y)
    (set-box! y 10)))

> (foo-by-ref (box 5))
10
```

Call-by-name

Note that 2 sets of ^^^ are printed in both call-by-name examples.

```
(define foo-by-name
  (lambda (th)
    (write (bar-by-name th))
    (newline)
    (th)))

(define bar-by-name
  (lambda (th^)
    (+ (th^) (th^))))

> (foo-by-name (lambda () (begin (write '^^^) (newline) (! 5))))
^^^
^^^
240
^^^
120
```

Another version of call-by-name, using boxes:

```
(define foo-by-name/boxes
  (lambda (bth)
    (write (bar-by-name/boxes bth))
    (newline)
    ((unbox bth))))

(define bar-by-name/boxes
  (lambda (bth^)
    (+ ((unbox bth^)) ((unbox bth^)))))

> (foo-by-name/boxes (box (lambda () (begin (write '^^^) (newline) (! 5)))))
^^^
^^^
240
^^^
120
```

Call-by-need

But in call-by-need, only one ^^^ is printed!

```
(define foo-by-need
  (lambda (bth)
    (write (bar-by-need bth))
    (newline)
    ((unbox bth))))

(define bar-by-need
   (lambda (bth^)
     (+ (let ((ans ((unbox bth^))))
          (set-box! bth^ (lambda () ans))
          ans)
        ((unbox bth^)))))
> (foo-by-need (box (lambda () (begin (write '^^^) (newline) (! 5)))))
^^^
240
120
```

Jensen's Device

Jensen's Device is a technique that relies on call-by-name. Here's an example:

```
(define baz
  (lambda (th)
    (+ (* (th) 5) (* (th) 5) (* (th) 5))))

> (let ((j 0))
    (baz (lambda () (begin (set! j (+ j 1)) j))))
30
```

Note that this will not work in any other call-by, since it relies on evaluating the side effect each time.

call-by-notes.txt · Last modified: 2013/03/17 00:16 by menzel